

## Die ersten Schritte mit Pablo

### **Aufgabenbeschreibung:**

Die erste Programmieraufgabe dient dazu, die verschiedenen Bewegungen des Roboters kennenzulernen. Dabei wird die allgemeine Handhabung sowie die Bedienung des smartPads geübt. Während dieser Übung wird der Roboter einen Stift aufnehmen und verschiedene Bewegungsarten abfahren. Diese Bewegungsarten werden mithilfe des Stiftes auf einer Papiervorlage festgehalten. Es werden hierbei die Bewegungen **LIN**, **CIRC** und **PTP** einmal mit und einmal ohne den Einsatz des Überschleifens abgefahren. So können mit der Übung Erkenntnisse über die verschiedenen Bewegungsarten festgehalten werden, welche für die späteren Übungen sinnvoll sein können.

### **Ziele der Übung:**

- Erste Bewegungen mit dem Roboter durchführen
- Satzanwahl und Teachen
- Roboter im Handbetrieb T1 fahren
- Kartesisches Achsenverfahren testen
- POV und HOV Geschwindigkeit einstellen
- Erkenntnisse über die einzelnen Bewegungen gewinnen
- Unterschied zwischen Bewegung mit und ohne Überschleifen
- Bedienung des smartPads erlernen

### **Aufbau der Übung:**

Für die erste Übung mit dem Roboter benötigt jeder Teilnehmer die auf der Moodle Plattform bereitgestellte Zeichenvorlage. Diese wird mit Klebeband auf dem Werkstisch des Roboters befestigt. Die Stelle, an dem die Vorlage platziert wird kann beliebig sein. Es sollte jedoch darauf geachtet werden, dass die Vorlage weder zu dicht dran noch zu weit weg vom Roboter platziert ist, damit dieser die Vorlage noch erreichen kann. Die folgende Abbildung 1 dient als Orientierung zur Platzierung der Vorlage.




Abbildung 1: Aufbau der Übung 1

Für die Übung wird der Filzstift in der Halterung benötigt. Vor dem Versuchsbeginn sollte überprüft werden, ob der Stift noch funktionstüchtig ist. Da der Filzstift während der Programmierung der Aufgabe teilweise länger auf der Vorlage abgesetzt wird, kann es dazu kommen, dass die Farbe durchdrückt. Es sollte daher unter die Vorlage noch 1-2 Seiten Schmierpapier gelegt werden, um ein durchdrücken zu verhindern.

**Durchführung der Übung:**

Für diese Übung existiert ein bereits implementiertes Programm „Uebung1“. Dieses sollte nach der Dearchivierung des Roboterprojektes (wie bereits in der Vorbereitung geschehen) identisch mit dem Programm sein, welches in den nachfolgenden Abbildungen gezeigt ist.

```

1 DEF Uebung1( )
2
3
4 ; *****
5 ; *****
6 ; *      BEWEGUNGSARTEN      *
7 ; *****
8 ; *****
9
10 →INI
11
12 ; *****
13 ; *      Stiftaufnahme      *
14 ; *****
15 PTP HOME Vel=100 % DEFAULT
16 PTP P31 CONT Vel=50 % PDAT24 Tool[1] Base[0]
17 LIN P32 Vel=0.5 m/s CPDAT11 Tool[1] Base[0]
18 WAIT FOR ( IN 4 'Steuerluft i.O.' ) AND ( IN 3
  ↳ 'Keine Zutrittsanforderung' ) CONT
19 SYN PULSE 258 'Greifer1 Schliessen' State=TRUE
  ↳ Time=0.1 sec at START Delay=0 ms
20 LIN P33 Vel=0.5 m/s CPDAT12 Tool[1] Base[0]
21
22 ; *****
23 ; *      Ohne Ueberschleifen      *
24 ; *****
25 PTP HOME Vel= 100 % DEFAULT
26 PTP P10 CONT Vel=90 % PDAT14 Tool[1] Base[0]
27
28 ; ----- LIN Bewegung -----
29 PTP P1 Vel=50 % PDAT4 Tool[1] Base[0]
30 LIN P2 Vel=0.5 m/s CPDAT3 Tool[1] Base[0]
31 LIN P3 Vel=0.5 m/s CPDAT4 Tool[1] Base[0]
32 PTP P4 CONT Vel=100 % PDAT5 Tool[1] Base[0]
33 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
34
35 ; ----- PTP Bewegung -----
36 PTP P1 Vel=50 % PDAT6 Tool[1] Base[0]

```

Abbildung 2: Uebung 1 (1)

```

37 PTP P2 Vel=50 % PDAT7 Tool[1] Base[0]
38 PTP P3 Vel=50 % PDAT8 Tool[1] Base[0]
39 PTP P4 Vel=100 % PDAT9 Tool[1] Base[0]
40 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
41
42 ; -----CIRC Bewegung -----
43 PTP P1 Vel=50 % PDAT10 Tool[1] Base[0]
44 CIRC P2 P3 Vel=0.5 m/s CPDAT7 Tool[1] Base[0]
45 PTP P4 CONT Vel=50 % PDAT13 Tool[1] Base[0]
46 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
47
48
49 ; *****
50 ; *      Mit Ueberschleifen      *
51 ; *****
52 PTP HOME Vel=100 % DEFAULT
53 PTP P10 CONT Vel=50 % PDAT15 Tool[1] Base[0]
54 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
55
56 ; ----- LIN Bewegung -----
57 PTP P1 CONT Vel=50 % PDAT16 Tool[1] Base[0]
58 LIN P2 CONT Vel=0.5 m/s CPDAT8 Tool[1] Base[0]
59 LIN P3 Vel=0.5 m/s CPDAT9 Tool[1] Base[0]
60 PTP P4 CONT Vel=100 % PDAT17 Tool[1] Base[0]
61 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
62
63 ; ----- PTP Bewegung -----
64 PTP P1 CONT Vel=50 % PDAT18 Tool[1] Base[0]
65 PTP P2 CONT Vel=50 % PDAT19 Tool[1] Base[0]
66 PTP P3 Vel=50 % PDAT20 Tool[1] Base[0]
67 PTP P4 CONT Vel=100 % PDAT21 Tool[1] Base[0]
68 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
69
70 ; -----CIRC Bewegung -----
71 PTP P1 CONT Vel=50 % PDAT22 Tool[1] Base[0]
72 CIRC P2 P3 CONT Vel=0.5 m/s CPDAT10 Tool[1] Base[0]
73 PTP P4 CONT Vel=50 % PDAT23 Tool[1] Base[0]

```

Abbildung 3: Uebung 1 (2)

```

74 PTP HOME Vel=100 % DEFAULT
75 WAIT FOR ( IN 3 'Keine Zutrittsanforderung' ) CONT
76
77 ; *****
78 ; *      Stiftablage      *
79 ; *****
80 PTP P34 CONT Vel=50 % PDAT25 Tool[1] Base[0]
81 LIN P35 Vel=0.1 m/s CPDAT13 Tool[1] Base[0]
82 WAIT FOR ( IN 4 'Steuerluft i.O.' ) AND ( IN 3
  ↳ 'Keine Zutrittsanforderung' ) CONT
83 SYN PULSE 261 'Greifer1 Oeffnen' State=TRUE
  ↳ Time=0.1 sec at START Delay=0 ms
84 LIN P36 Vel=0.5 m/s CPDAT14 Tool[1] Base[0]
85 PTP HOME Vel=100 % DEFAULT
86
87 END

```

Abbildung 4: Uebung 1 (3)

Das Programm besteht aus mehreren Teilen. Es wird zuerst der Stift aus der Halterung aufgenommen und danach fährt der Roboter wieder in die Grundstellung. Anschließend beginnen die Bewegungsabfolgen mit den verschiedenen Bewegungsarten, erst ohne Überschleifen und nach einer erneuten Fahrt in die Grundstellung mit Überschleifen. Abschließend legt der Roboter den Stift wieder in der Ablage ab.

**WICHTIG: Die Programmzeilen, welche die Stiftaufnahme und Stiftablage dürfen NICHT verändert werden, da es sonst zu Kollisionen und Beschädigungen der Materialien kommen kann!**

Nachdem der Roboter gestartet, depassiviert und das Programm am smartPad **angewählt** wurde, kann die Übung 1 beginnen. Stellen Sie den Roboter auf das kartesische Achsenverfahren ein, indem Sie das Koordinatensystem **WELT/WORLD** auswählen. Die gesamte Übung soll ausschließlich mit diesem Koordinatensystem durchgeführt werden. Nachdem Sie die Geschwindigkeiten eingestellt haben (**POV: 10, HOV: 10**), können Sie im Handbetrieb **T1** das Programm bis einschließlich der Zeile **25** ausführen. Das Erreichen der Zeile 25 wird durch den Pfeil → angezeigt. Wenn die Zeile verfehlt wurde, kann über den „Start Rückwärts Taster“ das Programm zurückgefahren werden.

Jetzt müsste sich der Roboter in der Grundposition befinden und den Stift mit seinem Greifer aufgenommen haben. Der folgende Programmablauf stützt sich ab hier nur noch auf fünf verschiedene Punkt (P10, P1, P2, P3, P4). Diese sollen in dieser Übung neu **geteacht** werden.

Die Punkte P1 bis P4 sind dabei auf dem Blatt gekennzeichnet und sollen mit dem Roboter im Handbetrieb angefahren und anschließend geteacht werden. Dafür wird der Roboter nacheinander zu den einzelnen Punkten gefahren. Anschließend wird über die Satzanwahl eine Programmzeile angewählt, wo der entsprechende Punkt verwendet wird (bei P1 z.B. Zeile 29). Dann wird mittels des Tasters **Touch-Up** der aktuelle Punkt gespeichert. Dabei werden **alle** Punkte, welche P1 heißen, überschrieben! (Das Pop-Up Fenster nach dem Touch Up mit **JA** bestätigen). Der Touch-Up-Vorgang ist daher nur **einmal für jeden Punkt** notwendig.

Sie müssen nur die Punkt teachen und nicht die Wege! Nachdem beispielsweise der Punkt P1 geteacht wurde, fahren Sie den Roboter wieder nach oben, über eine Z-Achsenbewegung. Anschließend fahren Sie über die X-, Y-, und Z-Achse zum Punkt P2 und teachen diesen. Achten Sie bei den Punkten P1-P3 darauf dass die Miene des Stiftes **ungefähr zu Hälfte** eingedrückt ist. Es handelt sich um einen Filzstift mit flexibler Mine, daher ist dieses unbedenklich. Nachdem die Punkte P1-P4 geteacht sind, kann noch der Punkt P10 folgen. Dieser sollte ungefähr 20 cm über Punkt P1 liegen.

Wenn alle Punkte geteacht wurden, verfahren Sie den Roboter über eine Satzanwahl in Zeile **25** wieder in die Grundstellung. (Nach der Satzanwahl wird der Roboter mittels einer Zustimmungstaste und der Start-Taste verfahren) Achten Sie darauf, dass es bei dieser **Fahrt nicht zu Kollisionen kommt!**

Nun können Sie die Geschwindigkeit **POV** auf **30** stellen und den Programmablauf testen. Der Roboter sollte nun die Vorlage „bemalen“. Kommt es während des Durchlaufs zu Komplikationen, stoppen Sie den Roboter **sofort**, indem Sie die Zustimmungstaste loslassen! Drosseln Sie die Geschwindigkeiten bei Passagen, die Ihnen kritisch erscheinen! Bevor der Roboter nach erfolgreichem Durchlauf zum Programmteil „Stiftablage“ kommt, drosseln Sie die Geschwindigkeit **POV auf 10**.

**Falls während es Programms der Stift verrutscht ist bzw. nicht mehr so im Greifer fixiert ist wie nach der Stiftaufnahme, muss der Roboter gestoppt werden. In diesem Fall müssen Sie den Greifer manuell mit den Statustasten öffnen (siehe Hinweise) und das Programm ohne den Stift weiterlaufen lassen! Den Stift können Sie anschließend nach Beendigung des Programms wieder in die Halterung stellen.**

**Hinweise und Tipps zur Übung:**

- Diese Übung wird mit geschlossenem Schutzgitter durchgeführt. Dieses muss über den Taster „**Freigabe Schutzbereich**“ quittiert werden. **NUR** wenn die Lehrkraft ausschließlich die Überbrückung des Schutzkreises erlaubt, darf auch innerhalb des Schutzkreises gearbeitet werden. Für die Überbrückung des Schutzkreises ist der Schaltschranckschlüssel notwendig.
- Nutzen Sie bei dieser Übung nur den Button „**Touch Up**“. Die weiteren Parameter der Inline Funktionen dürfen nicht verändert werden.
- Es müssen insgesamt nur fünf Punkte angefahren werden. Die Reihenfolge ist dabei nicht relevant. Achten Sie darauf, dass Sie den Stift beim Teachen nur dreimal (bei den Punkten P1 bis P3) aufsetzen. Die Wege zwischen den Punkten soll der Roboter „in der Luft“ zurücklegen!
- Verwenden Sie nur den Handbetrieb **T1!**
- Einstellungen der Geschwindigkeit während des „Malens“:
  - **POV: 30**
  - **HOV: 10**
  - Bei schwierigen, engen Passagen sollte die **Geschwindigkeit angemessen gedrosselt** werden
- Es darf ausschließlich das **kartesische Achsenverfahren** genutzt werden!
- Nutzen Sie die **Statustaste** , um mittels der **Gripper&SpotTech** Software den Greifer zu öffnen oder zu schließen, wenn es Probleme mit dem Stift gibt. (Zustimmungstaste erforderlich)
- **ACHTUNG:** Versuchen Sie **nie** den Stift per Hand aus dem Greifer zu nehmen! Benutzen Sie dafür immer die Statustasten des Roboters!
- Das Programm muss **angewählt** und nicht bloß geöffnet sein!